


Multi-Discipline Finite Element Analysis



Applications of FlexPDE

Volume 1 revision 1

Electricity and Magnetism

Introduction

FlexPDE

FlexPDE is a software tool for finding numerical solutions to systems of linear or non-linear partial differential equations using the methods of finite element analysis. The systems may represent static boundary value, time dependent initial/boundary value, or eigenvalue problems. Rather than addressing the solution of specific equations related to a given area of application, FlexPDE provides a framework for treating partial differential equation systems in general. It gives users a straightforward method of defining the equations, domains and boundary conditions appropriate to their application. From this description it creates a finite element solution process tailored to the problem. Within quite broad limits, then, FlexPDE is able to construct a numerical solution to a wide range of applications, without itself having any built-in knowledge of any of them.

The goal of this book is *not* to provide a discussion of the specific grammatical rules of writing scripts for FlexPDE, *nor* to describe the operation of the graphical user interface. Those topics are covered in other volumes of the FlexPDE documentation, the Getting Started guide, the User Guide tutorial, and the Problem Descriptor Reference.

In this book we will address several fields of physics in which FlexPDE finds fruitful application, describing the various problems, the mathematical statement of the partial differential equation system, and the ultimate posing of the problem to FlexPDE. The volume is accompanied by the text of all the examples, which the user can submit to FlexPDE to see the solution in progress or use as a foundation for problems of his own.

This manual is emphatically not a compendium of the problems FlexPDE “knows how to solve”. It is rather a group of examples showing ways in which the power of FlexPDE can be applied to partial differential equations systems in many fields. The true range of applicability of FlexPDE can be demonstrated only by the full range of ingenuity of users with insight into the mathematics of their own special fields.

Nor does this manual attempt to present textbook coverage of the theory of the topics addressed. The range of applications addressable by FlexPDE would make such an attempt impossible, even if we were capable of such an endeavor. Instead, we have presented enough of the theory of each topic to allow those practitioners who are familiar with the subject to see how the material has been analyzed and presented to FlexPDE. Users who are unfamiliar with the various fields

of application should consult standard textbooks to find the full theoretical development of the subjects.

Finite Element Methods

It is not our intent to provide an elaborate discussion of finite element methods. One goal of FlexPDE has been to allow users in the various fields of science and engineering to begin reaping the benefits of applying finite element analysis to their individual work without becoming programmers and numerical analysts. There are hundreds of books in print detailing the method and its variants in many fields, and the interested student can find a wealth of material to keep him busy. If we have been successful in our endeavors, he won't have to.

Nevertheless, a familiarity with some of the concepts of finite element analysis can be of benefit in understanding how FlexPDE works, and why it sometimes doesn't. Hence this brief overview.

Principles

Partial differential equations generally arise as a mathematical expression of some conservation principle such as a conservation of energy, momentum or mass. Partial differential equations by their very nature deal with continuous functions -- a derivative is the result of the limiting process of observing differences at an infinitesimal scale. A temperature distribution in a material, for example, is assumed to vary smoothly between one extreme and another, so that as we look ever more closely at the differences between neighboring points, the values become ever closer until at "zero" separation, they are the same.

Computers, on the other hand, apply arithmetic operations to discrete numbers, of which only a limited number can be stored or processed in finite time. A computer cannot analyze an infinitude of values. How then can we use a computer to solve a real problem?

Many approaches have been devised for using computers to approximate the behavior of real systems. The finite element method is one of them. It has achieved considerable success in its few decades of existence, first in structural mechanics, and later in other fields. Part of its success lies in the fact that it approaches the analysis in the framework of integrals over small patches of the total domain, thus enforcing aggregate correctness even in the presence of microscopic error. The techniques applied are little dependent on shapes of objects, and are therefore applicable in real problems of complex configuration.

The fundamental assumption is that no matter what the shape of a solution might be over the entire domain of a problem, at some scale each local patch of the solution can be well approximated by a low-

order polynomial. This is closely related to the well-known Taylor series expansion, which expresses the local behavior of a function in a few polynomial terms.

In a two-dimensional heat flow problem, for example, we assume that if we divide the domain up into a large number of triangular patches, then in each patch the temperature can be well represented by, let us say, paraboloidal surfaces. Stitching the patches together, we get a Harlequin surface that obeys the differential limiting assumption of continuity for the solution value—but perhaps not for its derivatives. The patchwork of triangles is referred to as the computation "mesh", and the sample points at vertices or elsewhere are referred to as the "nodes" of the mesh.

In three dimensions, the process is analogous, using a tetrahedral subdivision of the domain.

How do we determine the shape of the approximating patches?

1. Assign a sample value to each vertex of the triangular or tetrahedral subdivision of the domain. Then each vertex value is shared by several triangles (tetrahedra).
2. Substitute the approximating functions into the partial differential equation.
3. Multiply the result by an importance-weighting function and integrate over the triangles surrounding each vertex.
4. Solve for the vertex values which minimize the error in each integral.

This process, known as a "weighted residual" method, effectively converts the continuous PDE problem into a discrete minimization problem on the vertex values. This is usually known as a "weak form" of the equation, because it does not strictly enforce the PDE at all points of the domain, but is instead correct in an integral sense relative to the triangular subdivision of the domain.

The locations and number of sample values is different for different interpolation systems. In FlexPDE, we use either quadratic interpolation (with sample values at vertices and midsides of the triangular cells), or cubic interpolation (with values at vertices and two points along each side). Other configurations are possible, which gives rise to various "flavors" of finite element methods.

Boundary Conditions

A fundamental component of any partial differential equation system is the set of boundary conditions, which alone make the solution unique. The boundary conditions are analogous to the integration constants

that arise in integral calculus. We say $\int x^2 dx = \frac{1}{3}x^3 + C$, where C is any constant. If we differentiate the right hand side, we recover the integrand, regardless of the value of C .

In a similar way, to solve the equation $\frac{\partial^2 u}{\partial x^2} = 0$, we must integrate

twice. The first integration gives $\frac{\partial u}{\partial x} + C_1$, and the second

gives $C_1 x + C_2$. These integration constants must be supplied by the boundary conditions of the problem statement.

It is clear from this example that there are as many integration constants as there are nested differentiations in the PDE. In the general case, these constants can be provided by a value at each end of an interval, a value and a derivative at one end, etc. In practice, the most common usage is to provide either a value or a derivative at each end of the domain interval. In two or three dimensions, a value or derivative condition applied over the entire bounding curve or surface provides one condition at each end of any coordinate integration path.

Integration by Parts and Natural Boundary Conditions

A fundamental technique applied by FlexPDE in treating the finite element equations is “integration by parts”, which reduces the order of a derivative integrand, and also leads immediately to a formulation of derivative boundary conditions for the PDE system.

In its usual form, integration by parts is given as

$$\int_a^b u dv = (uv) \Big|_a^b - \int_a^b v du .$$

Application of integration by parts to a vector divergence in a two- or three-dimensional domain, for example, results in the Divergence Theorem, given in 2D as

$$\iint_A \nabla \cdot \vec{F} dA = \oint_l \vec{F} \cdot \hat{n} dl .$$

This equation relates the integral inside the area to the flux crossing the outer boundary (\hat{n} referring to the outward surface-normal unit vector).

As we shall see, the use of integration by parts has a wide impact on the way FlexPDE interprets and solves PDE systems.

Applied to the weighted residual method, this process dictates the flux conservation characteristics of the finite element approximation at boundaries between the triangular approximation cells, and also provides a method for defining the interaction of the system with the outside world, by specifying the value of the surface integrand.

The values of the surface integrands are the “Natural” boundary conditions of the PDE system, a term which also arises in a similar context in variational calculus.

FlexPDE uses the term “Natural” boundary condition to specify the boundary flux terms arising from the integration by parts of all second-order terms in the PDE system.

For example, in a heat equation, $\nabla \cdot (-k \nabla \varphi) + S = 0$, the divergence term will be integrated by parts, resulting in

$$(0.1) \quad \iint_A \nabla \cdot (-k \nabla \varphi) dA = \oint_l (-k \nabla \varphi) \cdot \hat{n} dl$$

The right hand side is the heat flux crossing the outer boundary, and the value of $-k \nabla \varphi$ must be provided by the user in a Natural boundary condition statement (unless a value BC is applied instead).

At an interface between two materials, $-k_1 (\nabla \varphi)_1 \cdot \hat{n}_1$ represents the heat energy leaving material 1 at a point on the interface. Likewise, $-k_2 (\nabla \varphi)_2 \cdot \hat{n}_2$ represents the heat energy leaving material 2 at the same point. Since the outward normal from material 1 is the negative of the outward normal from material 2, the sum of the fluxes at the boundary is $[k_2 (\nabla \varphi)_2 - k_1 (\nabla \varphi)_1] \cdot \hat{n}_1$, and this becomes the Natural boundary condition at the interface. In this application, we want energy to be conserved, so that the two flux terms must sum to zero. Thus the internal Natural BC is zero at the interface, and this is the default value applied by FlexPDE.

Useful Integral Rules

$$(0.2) \quad \iiint_V \nabla f dV = \iint_S (\vec{n}f) dS \quad (\text{Gradient Theorem})$$

$$(0.3) \quad \iiint_V \nabla \cdot \vec{F} dV = \iint_S (\vec{n} \cdot \vec{F}) dS \quad (\text{Divergence Theorem})$$

$$(0.4) \quad \iiint_V \varphi \nabla \cdot \vec{F} dV = \iint_S \varphi (\vec{n} \cdot \vec{F}) dS - \iiint_V (\nabla \varphi) \cdot \vec{F} dV$$

$$(0.5) \quad \iiint_V \nabla \times \vec{F} dV = \iint_S (\vec{n} \times \vec{F}) dS \quad (\text{Curl Theorem})$$

Adaptive Mesh Refinement

We have said that at “some scale”, the solution can be adequately approximated by a set of low-order polynomials. But it is not always obvious where the mesh must be dense and where a coarse mesh will suffice. In order to address this issue, FlexPDE uses a method of “adaptive mesh refinement”. The problem domain presented by the user is divided into a triangular mesh dictated by the feature sizes of the domain and the input controls provided by the user. The problem is then constructed and solved, and the cell integrals of the weighted residual method are crosschecked to estimate their accuracy. In locations where the integrals are deemed to be of questionable accuracy, the triangles are subdivided to give a new denser mesh, and the problem is solved again. This process continues until FlexPDE is satisfied that the approximation is locally accurate to the tolerance assigned by the user. Acceptable local accuracy does not necessarily guarantee absolute accuracy, however. Depending on how errors accumulate or cancel, the global accuracy could be better or worse than the local accuracy condition implies.

Time Integration

The finite element method described above is most successful in treating boundary value problems. When addressing initial value problems, while the finite element method could be applied (and sometimes is), other techniques are frequently preferable. FlexPDE uses a variable-order implicit backward difference method (BDM) as introduced by C.W. Gear. In most cases, second order gives the best tradeoff between stability, smoothness and speed, and this is the default configuration for FlexPDE. This method fits a quadratic in time to each nodal value, using two known values and one future (unknown) value. It then solves the coupled equations for the array of nodal values at the new time. By looking backward one additional step, it is possible to infer the size of the cubic term in a four-point expansion of the time behavior of each nodal value. If these cubic contributions are

large, the timestep is reduced, and if extreme, the current step repeated.

Summary

With this very cursory examination of finite element methods, we are ready to start applying FlexPDE to the solution of PDE systems of interest in real scientific and engineering work.

Disclaimer

We have tried to make these notes as accurate as possible, but because we are not experts in all the fields addressed, it is possible that errors have crept in. We invite readers to comment freely on the material presented here, and to take us to task if we have erred.

Chapter 1

Electrostatics

Perhaps the most important of all partial differential equations is the simple form

$$(1.1) \quad \nabla \cdot (k \nabla \varphi) + q = 0$$

It is encountered in virtually all branches of science and engineering, and describes the diffusion of a quantity φ with diffusivity k and volume source q . With $k=1$ it is referred to as Poisson's equation,

$\nabla^2 \varphi + q = 0$. With $k=1$ and $q=0$, it is referred to as Laplace's equation, $\nabla^2 \varphi = 0$.

If φ is electric potential, k is permittivity and q is charge density, then (1.1) is the electrostatic field equation.

If φ is temperature, k is thermal conductivity and q is heat source, then (1.1) is the heat equation.

If we identify derivatives of φ with fluid velocities,

$$u = \frac{\partial \varphi}{\partial x}, \quad v = \frac{\partial \varphi}{\partial y}$$

then (1.1) is the potential flow equation.

In most cases, we can identify $-k \nabla \varphi$ with the flux of some quantity such as heat, mass or a chemical. (1.1) then says that the variation of the rate of transfer of the relevant quantity is equal to the local source (or sink) of the quantity.

If we integrate the divergence term by parts (or equivalently, apply the divergence theorem), we get

$$(1.2) \quad \iiint_V \nabla \cdot (k \nabla \varphi) dV = \iint_S \vec{n} \cdot (k \nabla \varphi) dS = - \iiint_V q dV$$

That is, the total interior source is equal to the net flow across the outer boundary.

In a FlexPDE script, the equation (1.1) is represented simply as

```
Div(k*grad(phi)) + q = 0
```

The boundary flow $\vec{n} \cdot (k \nabla \varphi)$ is represented in FlexPDE by the Natural boundary condition,

```
Natural(phi) = <boundary flux>
```

The simplest form of the natural boundary condition is the insulating or "no flow" boundary,

```
Natural(phi) = 0.
```

Electrostatic Fields in 2D

Let us as a first example construct the electrostatic field equation for an irregularly shaped block of high-dielectric material suspended in a low-dielectric material between two charged plates.

First we must present a title:

```
title
'Electrostatic Potential'
```

Next, we must name the variables in our problem:

```
variables
v
```

We will need the value of the permittivity:


```
definitions
eps = 1
```

The equation is as presented above, using the div and grad operators in place of $\nabla \cdot$ and ∇ :

```
equations
div(eps*grad(v)) = 0
```

The domain will consist of two regions; the bounding box containing the entire space of the problem, with charged plates top and bottom:

```
boundaries
region 1
start (0,0)
value(v) = 0
line to (1,0)
natural(v) = 0
line to (1,1)
value(v) = 100
```



```

line to (0,1)
natural(V) = 0
line to finish


```

and the imbedded dielectric:

```

region 2
eps = 50
start (0.4,0.4)
line to (0.8,0.4)
to (0.8,0.8)
to (0.6,0.8)
to (0.6,0.6)
to (0.4,0.6)
to finish

```



Notice that we have used the insulating form of the natural boundary condition on the sides of the bounding box, with specified potentials top (100) and bottom (0).

We have specified a permittivity of 50 in the imbedded region. (Since we are free to multiply through the equation by the free-space permittivity ϵ_0 , we can interpret the value as relative permittivity or dielectric constant.)

What will happen at the boundary between the dielectric and the air? If we apply equation (1.2) and integrate around the dielectric, we get

$$\oint_l \vec{n} \cdot (k \nabla \phi) dl = \iint_A q dA = 0$$

If we perform this integration just inside the boundary of the dielectric, we must use $k = 50$, whereas just outside the boundary, we must use $k = 1$. Yet both integrals must yield the same result. It therefore follows that the interface condition at the boundary of the dielectric is

$$\vec{n} \cdot (k \nabla \phi)_{\text{inside}} = \vec{n} \cdot (k \nabla \phi)_{\text{outside}} .$$

Since the electric field vector is $\vec{E} = -\nabla \phi$ and the electric displacement is $\vec{D} = \epsilon \vec{E}$, we have the condition that the normal component of the electric displacement is continuous across the interface, as required by Maxwell's equations.

We want to see what is happening while the problem is being solved, so we add a monitor of the potential:

```

monitors
contour(V) as 'Potential'

```

At the end of the problem we would like to save as graphical output the computation mesh, a contour plot of the potential, and a vector plot of the electric field:

```

plots
grid(x,y)
contour(V) as 'Potential'
vector(-dx(V),-dy(V)) as 'Electric Field'

```

The problem specification is complete, so we end the script:

```

end

```

Putting all these sections together, we have the complete script for the dielectric problem:




Descriptor 1.1: Dielectric.pde

```

title
'Electrostatic Potential'
variables
V
definitions
eps = 1
equations
div(eps*grad(V)) = 0
boundaries
region 1
start (0,0)
value(V) = 0 line to (1,0)
natural(V) = 0 line to (1,1)
value(V) = 100 line to (0,1)
natural(V) = 0 line to finish
region 2
eps = 50
start (0.4,0.4)
line to (0.8,0.4) to (0.8,0.8)
to (0.6,0.8) to (0.6,0.6)
to (0.4,0.6) to finish
monitors
contour(V) as 'Potential'
plots
grid(x,y)
contour(V) as 'Potential'
vector(-dx(V),-dy(V)) as 'Electric Field'
end

```

The output plots from running this script are as follows:



Electrostatics in 3D

We can convert this example quite simply to a three dimensional calculation. The modifications that must be made are:

- Specify cartesian3 coordinates.
- Add an extrusion section listing the dividing surfaces.
- Provide boundary conditions for the end faces.
- Qualify plot commands with the cut plane in which the plot is to be computed.

In the following descriptor, we have divided the extrusion into three layers. The dielectric constant in the first and third layer are left at the default of $k=1$, while layer 2 is given a dielectric constant of 50 in the dielectric region only.

A contour plot of the potential in the plane $x=0$ has been added, to show the resulting vertical cross section. The plots in the $z=0.15$ plane reproduce the plots shown above for the 2D case.

Modifications to the 2D descriptor are shown in red.

Descriptor 1.2: 3D Dielectric.pde

title

```

'Electrostatic Potential'

coordinates
  cartesian3

variables
  V
definitions
  eps = 1
equations
  div(eps*grad(V)) = 0

extrusion
  surface "bottom" z=0
  surface "dielectric_bottom" z=0.1
  layer "dielectric"
  surface "dielectric_top" z=0.2
  surface "top" z=0.3

boundaries

  surface "bottom" natural(V)=0
  surface "top" natural(V)=0

region 1
  start (0,0)
  value(V) = 0      line to (1,0)
  natural(V) = 0     line to (1,1)
  value(V) = 100     line to (0,1)
  natural(V) = 0     line to finish

region 2
  layer "dielectric" eps = 50
  start (0.4,0.4)
  line to (0.8,0.4) to (0.8,0.8)
  to (0.6,0.8) to (0.6,0.6)
  to (0.4,0.6) to finish


monitors
  contour(V) on z=0.15 as 'Potential'

plots
  contour(V) on z=0.15 as 'Potential'
  vector(-dx(V),-dy(V)) on z=0.15
  as 'Electric Field'
  contour(V) on x=0.5 as 'Potential'

end

```

The following potential plot on $x=0$ shows the vertical cross section of the extruded domain. Notice that the potential pattern is not symmetric, due to the influence of the extended leg of the dielectric in the y direction.



Capacitance per Unit Length in 2D Geometry¹

This problem illustrates the calculation of capacitance per unit length in a 2D X-Y geometry extended indefinitely in the Z direction. The capacitance is that between a conductor enclosed in a dielectric sheath and a surrounding conductive enclosure. In addition to these elements, there is also another conductor (also with a dielectric sheath) that is "free floating" so that it maintains zero net charge and assumes a potential that is consistent with that uncharged state.

We use the potential V as the system variable, from which we can calculate the electric field $\vec{E} = \nabla V$ and displacement $\vec{D} = \epsilon \vec{E}$, where ϵ is the local permittivity and may vary with position.

In steady state, in charge-free regions, Maxwell's equation then becomes

$$\nabla \cdot \vec{D} = \nabla \cdot (\epsilon \vec{E}) = \nabla \cdot (\epsilon \nabla V) = 0.$$

We impose value boundary conditions on V at the surfaces of the two conductors, so that we do not have to deal with regions that contain charge.

The metal in the floating conductor is "faked" with a fairly high permittivity, which has the effect of driving the interior field and field energy to near zero. The imposition of (default) natural boundary conditions then keeps the field normal to the surface of the conductor, as Maxwell requires. Thus we get a good answer without having to solve for the charge on the floating conductor, which would be a real pain due to its localization on the surface of the conductor.

The capacitance can be found in two ways. If we know the charge Q on the conductor at fixed potential V , we solve $Q = CV$ to get $C = Q/V$. We know V because it is imposed as a boundary condition, and we can find Q from the fact that

$$\oint_s \vec{n} \cdot \vec{D} = Q$$

where the integral is taken over a surface enclosing a volume and Q is the charge in the volume.

Alternatively, we can use the energy relation $W = \frac{1}{2} CV^2$ to get

$C = 2W/V^2$. We find the energy W by integrating the energy density $\frac{1}{2} \vec{E} \cdot \vec{D}$ over the area of the problem.

Descriptor 1.3: Capacitance.pde

```
TITLE 'Capacitance per Unit Length of 2D Geometry'
{ 17 Nov 2000 by John Trenholme }

SELECT
  errlim 1e-4
  thermal_colors on
  plotintegrate off

VARIABLES
  v

DEFINITIONS
  mm = 0.001                ! meters per millimeter
  Lx = 300 * mm             ! enclosing box dimensions
  Ly = 150 * mm
  b = 0.7                    ! fractional radius of conductor
  ! position and size of cable at fixed potential:
  x0 = 0.25 * Lx
  y0 = 0.5 * Ly
  r0 = 15 * mm
  x1 = 0.9 * Lx
  y1 = 0.3 * Ly
  r1 = r0
  epsr                ! relative permittivity
  epsd = 3             ! epsr of cable dielectric
  epsmetal = 1000      ! fake metallic conductor
  eps0 = 8.854e-12     ! permittivity of free space
  eps = epsr * eps0
  v0 = 1                ! fixed potential of the cable

  ! field energy density:
  energyDensity = dot[ eps * grad( v), grad( v) ]/2

EQUATIONS
  div[ eps * grad( v) ] = 0

BOUNDARIES
  region 1 'inside' epsr = 1
  start 'outer' ( 0, 0) value( v) = 0
  line to (Lx,0) to (Lx,Ly) to (0,Ly) to finish
  region 2 'die10' epsr = epsd
  start 'dieb0' (x0+r0, y0)
  arc ( center = x0, y0) angle = 360
  region 3 'cond0' epsr = 1
  start 'conb0' (x0+b*r0, y0) value(v) = v0
  arc ( center = x0, y0) angle = 360
```

¹ Submitted by J.B. Trenholme

```

region 4 'diel1' epsr = epsd
start 'dieb1' ( x1+r1, y1)
arc ( center = x1, y1) angle = 360
region 5 'cond1' epsr = epsmetal
start 'conb1' ( x1+b*r1, y1)
arc ( center = x1, y1) angle = 360

```

PLOTS

```

contour( v) as 'Potential'
contour( v) as 'Potential Near Driven Conductor'
  zoom(x0-1.1*r0, y0-1.1*r0, 2.2*r0, 2.2*r0)
contour( v)
  as 'Potential Near Floating Conductor'
  zoom(x1-1.1*r1, y1-1.1*r1, 2.2*r1, 2.2*r1)
elevation( v) from ( 0,y0) to ( x0, y0)
  as 'Potential from Wall to Driven Conductor'
elevation( v) from ( x0, y0) to ( x1, y1)
  as 'Potential from Driven to Floating Conductor'
vector( grad( v)) as 'Field'
contour( energyDensity) as 'Field Energy Density'
contour( energyDensity)
  zoom( x1-1.2*r1, y1-1.2*r1, 2.4*r1, 2.4*r1)
  as 'Field Energy Density Near Floating Conductor'
elevation( energyDensity)
  from (x1-2*r1, y1) to ( x1+2*r1, y1)
  as 'Field Energy Density Near Floating Conductor'
contour( epsr) paint on "inside"
  as 'Definition of Inside'

```



SUMMARY



```

report sintegral(normal[eps*grad(v)], 'conb0', 'diel0')
  as 'Driven charge'
report sintegral(normal[eps*grad(v)], 'outer', 'inside')
  as 'Outer charge'
report sintegral(normal[eps*grad(v)], 'conb1', 'diel1')
  as 'Floating charge'
report sintegral(normal[eps*grad(v)], 'conb0', 'diel0')/v0
  as 'Capacitance (f/m)'
report integral( energyDensity, 'inside')
  as 'Energy (J/m)'
report 2 * integral( energyDensity, 'inside') / v0^2
  as 'Capacitance (f/m)'
report 2 * integral(energyDensity)/(v0*
  sintegral( normal[eps*grad(v)], 'conb0', 'diel0'))
  as 'cap_by_energy / cap_by_charge'

```

END





Chapter 2 Magnetostatics

From Maxwell's equations in a steady-state form we have

$$(2.1) \quad \nabla \times \vec{H} = \vec{J}$$

$$\nabla \cdot \vec{B} = 0$$

$$\nabla \cdot \vec{J} = 0$$

where \vec{H} is the magnetic field intensity, $\vec{B} = \mu \vec{H}$ is the magnetic induction, μ is the magnetic permeability and \vec{J} is the current density.

The conditions required by Maxwell's equations at a material interface are

$$(2.2) \quad \vec{n} \times \vec{H}_1 = \vec{n} \times \vec{H}_2$$

$$\vec{n} \cdot \vec{B}_1 = \vec{n} \cdot \vec{B}_2$$

It is sometimes fruitful to use the magnetic field quantities directly as variables in a model. However, eq. (2.2) shows that the tangential components of \vec{H} are continuous across an interface, while the normal components of \vec{B} are continuous.

The finite element method used by FlexPDE has a single value of each variable on an interface, and therefore requires that the quantities chosen for system variables must be continuous across the interface.

In special cases, it may be possible to choose components of \vec{B} or \vec{H} which satisfy this continuity requirement. We could, for example model B_x in a problem where material interfaces are normal to x . In the general case, however, meeting the continuity requirements can be impossible.

It is common in Magnetostatics to use instead of the field quantities the magnetic vector potential \vec{A} , defined as

$$(2.3) \quad \vec{B} = \nabla \times \vec{A}.$$

This definition automatically enforces $\nabla \cdot \vec{B} = 0$. Furthermore, \vec{A} can be shown to be continuous everywhere in the domain, and can represent the conditions (2.2) correctly.

\vec{A} can be derived from Ampere's Law, and shown to be the integrated effect at each point of all the current loops active in the domain. In this derivation, \vec{A} will have components parallel to the components of \vec{J} , so that it can be determined a priori which components of \vec{A} must be represented.

Eq. (2.3) alone is not sufficient to uniquely define \vec{A} . It must be supplemented by a definition of $\nabla \cdot \vec{A}$ to be unique. This definition (the "gauge condition") is usually taken to be $\nabla \cdot \vec{A} = 0$ ("Coulomb gauge"), a definition consistent with the derivation of \vec{A} from Ampere's Law. Other definitions are useful in some applications. It is not important what the gauge condition is; in all cases $\nabla \times \vec{A}$, and therefore the field quantities, remain the same.

Combining eq. (2.1) with (2.3) gives

$$(2.4) \quad \nabla \times ((\nabla \times \vec{A}) / \mu) = \vec{J}$$

In cases with multiple materials, where μ can take on different values, it is important to keep the μ inside the curl operator, because it is the integration of this term by parts that gives the correct jump conditions at the material interface.

Applying eq. (0.5) we have

$$(2.5) \quad \iiint_V \nabla \times ((\nabla \times \vec{A}) / \mu) dV = \iiint_V \nabla \times \vec{H} dV = \oint_S \vec{n} \times \vec{H} dS,$$

so that the Natural boundary condition defines $\vec{n} \times \vec{H}$ on external boundaries, and $\vec{n} \times \vec{H}$ is assumed continuous across internal boundaries, consistent with Maxwell's equations.

A Magnet Coil in 2D cylindrical coordinates

As a first example, we will calculate the magnetic field created by a coil, using 2D cylindrical (r,z) geometry. We will apply current only in the azimuthal direction, so the only nonzero component of \vec{A} will be the azimuthal component A_ϕ . With only a single component normal to the computational plane, the gauge condition is automatically satisfied,

$$\text{since } \nabla \cdot \vec{A} = \frac{1}{r} \frac{\partial A_\phi}{\partial \phi} = 0$$

In the descriptor which follows, note that we have chosen to align the cylindrical axis with the horizontal plot axis. FlexPDE uses a right-hand coordinate system, so in this case positive J_ϕ is outward from the plot page.

Descriptor 2.1: Magnet Coil.pde

```

title 'AXI-SYMMETRIC MAGNETIC FIELD'

coordinates
  xcylinder(Z,R)

Variables
  Aphi    { azimuthal component of the vector potential }

Definitions
  mu = 1                { the permeability }
  J = 0                 { global source term defaults to zero }
  current = 10          { the source value in the coil }
  Br = -dz(Aphi)        { definitions for plots }
  Bz = dr(r*Aphi)/r

Equations
  Curl(curl(Aphi)/mu) = J

Boundaries
  Region 1
    value(Aphi) = 0      { specify A=0 along axis }
    start(-10,0) line to (10,0)
    value(Aphi) = 0      { H x n = 0 on distant sphere }
    arc(center=0,0) angle 180 finish
  Region 2
    J = current          { redefine source value }
    start (-0.25,1)
    line to (0.25,1) to (0.25,1.5)
    to (-0.25,1.5) to finish

Monitors
  contour(Bz) zoom(-2,0,4,4) as 'FLUX DENSITY B'
  contour(Aphi) as 'Potential'

Plots

```




```

grid(z,r)
contour(Bz) as 'FLUX DENSITY B'
contour(Bz) zoom(-2,0,4,4) as 'FLUX DENSITY B'
elevation(Aphi, dr(Aphi), Aphi/r, Bz)
from (0,0) to (0,1) as 'Near Axis'
vector(Bz,Br) as 'FLUX DENSITY B'
vector(Bz,Br) zoom(-2,0,4,4) as 'FLUX DENSITY B'
contour(Aphi) as 'MAGNETIC POTENTIAL'

contour(Aphi) zoom(-2,0,4,4) as 'MAGNETIC POTENTIAL'
surface(Aphi) as 'MAGNETIC POTENTIAL'
viewpoint (-1,1,30)

```

End



Nonlinear Permeability in 2D

In the following 2D Cartesian example, a current-carrying copper coil is surrounded by a ferromagnetic core with an air gap. Current flows in the coil in the Z direction (out of the computation plane), and only the Z component of the magnetic vector potential is nonzero. The Coulomb gauge condition is again satisfied automatically. We assume a symmetry plane along the X-axis, and impose $A_z = 0$ along the remaining sides. The relative permeability is $\mu = 1$ in the air and the coil, while in the core it is given by

$$\mu = \frac{\mu_{\max}}{1 + C(\nabla A_z)^2} + \mu_{\min},$$

with parameters giving a behavior similar to transformer steel.

Descriptor 2.2: Saturation.pde

```

title "A MAGNETOSTATIC PROBLEM"

Select
  errlim = 1e-4

Variables
  A

Definitions
  mu = 1                { default to air }
  mu0 = 1               { for saturation plot }
  mu_max = 5000
  mu_min = 200
  mucore = mu_max / (1 + 0.05 * grad(A)**2) + mu_min
  S = 0
  current = 2
  y0 = 8

Equations
  curl(curl(A)/mu) = S

Boundaries
  Region 1              { The IRON core }
    mu = mucore
    mu0 = mu_max
    natural(A) = 0
    start(0,0)
      line to (40,0)
    value(A) = 0
    line to (40,40) to (0,40) finish
  Region 2              { The AIR gap }
    mu = 1
    start (15,0)
      line to (40,0) to (40,y0) to (32,y0)

```

```


      arc (center=32,y0+2) to (30,y0+2)
      line to (30,20) to (15,20) to finish
  Region 3              { The COIL }
    S = current
    mu = 1
    start (15,12)
      line to (30,12) to (30,20) to (15,20) finish



Monitors
  contour(A)

Plots
  grid(x,y)
  vector(dy(A), -dx(A)) as "FLUX DENSITY B"
  vector(dy(A)/mu, -dx(A)/mu) as "MAGNETIC FIELD H"
  contour(A) as "Az MAGNETIC POTENTIAL"
  surface(A) as "Az MAGNETIC POTENTIAL"
  contour(mu0/mu) painted as "Saturation: mu0/mu"

End

```





Divergence Form

In two dimensional geometry with a single nonzero component of \vec{A} , the gauge condition $\nabla \cdot \vec{A} = 0$ is automatically satisfied. Direct application of eq. (2.4) is therefore well posed, and we can proceed without further modification.

In 3D, however, direct implementation of eq. (2.4) does not impose a gauge condition, and is therefore ill-posed in many cases. One way to address this problem is to convert the equation to divergence form using the vector identity

$$(2.6) \quad \nabla \times (\nabla \times \vec{A}) = \nabla (\nabla \cdot \vec{A}) - \nabla^2 \vec{A}.$$

As long as μ is piecewise constant we can apply (2.6) together with the Coulomb gauge $\nabla \cdot \vec{A} = 0$ to rewrite (2.4) as

$$(2.7) \quad \nabla \cdot \left(\frac{\nabla \vec{A}}{\mu} \right) + \vec{J} = 0$$

If μ is variable, we can generalize eq (2.6) to the relation

$$(2.8) \quad \nabla \times \left(\frac{\nabla \times \vec{A}}{\mu} \right) = \nabla \cdot \left(\frac{\nabla \vec{A}}{\mu} \right)^T - \nabla \cdot \left(\frac{\nabla \vec{A}}{\mu} \right)$$

We assert without proof that there exists a gauge condition

$\nabla \cdot \vec{A} = F(x, y, z)$ which forces

$$(2.9) \quad \nabla \cdot \left(\frac{\nabla \vec{A}}{\mu} \right)^T = 0.$$

The equations governing F can be stated as

$$\frac{\partial}{\partial x} \left(\frac{F}{\mu} \right) = \frac{\partial}{\partial x} \left(\frac{1}{\mu} \frac{\partial A_y}{\partial y} + \frac{1}{\mu} \frac{\partial A_z}{\partial z} \right) - \frac{\partial}{\partial y} \left(\frac{1}{\mu} \frac{\partial A_y}{\partial x} \right) - \frac{\partial}{\partial z} \left(\frac{1}{\mu} \frac{\partial A_z}{\partial x} \right)$$

$$\frac{\partial}{\partial y} \left(\frac{F}{\mu} \right) = \frac{\partial}{\partial y} \left(\frac{1}{\mu} \frac{\partial A_x}{\partial x} + \frac{1}{\mu} \frac{\partial A_z}{\partial z} \right) - \frac{\partial}{\partial x} \left(\frac{1}{\mu} \frac{\partial A_x}{\partial y} \right) - \frac{\partial}{\partial z} \left(\frac{1}{\mu} \frac{\partial A_z}{\partial y} \right)$$

$$\frac{\partial}{\partial z} \left(\frac{F}{\mu} \right) = \frac{\partial}{\partial z} \left(\frac{1}{\mu} \frac{\partial A_x}{\partial x} + \frac{1}{\mu} \frac{\partial A_y}{\partial y} \right) - \frac{\partial}{\partial x} \left(\frac{1}{\mu} \frac{\partial A_x}{\partial z} \right) - \frac{\partial}{\partial y} \left(\frac{1}{\mu} \frac{\partial A_y}{\partial z} \right)$$

It is not necessary to solve these equations; we show them merely to indicate that F embodies the commutation characteristics of the system. The value of F is implied by the assertion (2.9). Clearly, when μ is constant, the equations reduce to $\nabla F = 0$, for which $F = 0$ is a solution.

Using the definition (2.9) we can again write the divergence form

$$(2.10) \quad \nabla \cdot \left(\nabla \vec{A} / \mu \right) + J = 0.$$

Boundary conditions

In converting the equation to a divergence, we have modified the interface conditions. The natural boundary condition for each component equation of (2.10) is now the normal component of the argument of the divergence:

$$(2.11) \quad \begin{aligned} \text{Natural}(A_x) &= \vec{n} \cdot \nabla A_x / \mu \\ \text{Natural}(A_y) &= \vec{n} \cdot \nabla A_y / \mu \\ \text{Natural}(A_z) &= \vec{n} \cdot \nabla A_z / \mu \end{aligned}$$

The default interior interface condition assumes component-wise continuity of the surface terms across the interface.

Of the conditions (2.2) required by Maxwell's equations at an interface, the first describes the tangential components of \vec{H} , which by (2.3) involve the normal components of $\nabla \vec{A}$. Eq. (2.11) shows that these components scale by $1/\mu$, satisfying the tangential condition on \vec{H} . The second condition is satisfied by the fact that the variables A_x, A_y, A_z have only a single representation on the boundary, requiring that their tangential derivatives, and therefore the normal component of \vec{B} , will be continuous across the interface.

In all cases it is important to keep the μ attached to the $\nabla \vec{A}$ term to preserve the correct interface jump conditions.

Magnetic Materials in 3D

In magnetic materials, we can modify the definition of \vec{H} to include magnetization and write

$$(2.12) \quad \vec{H} = \vec{B} / \mu - \vec{M}$$

We can still apply the divergence form in cases where $\vec{M} \neq 0$, but we must treat the magnetization terms specially.

The equation becomes:

$$(2.13) \quad \nabla \cdot \left(\frac{\nabla \vec{A}}{\mu} \right) + \nabla \times \vec{M} + \vec{J} = 0$$

FlexPDE does not integrate constant source terms by parts, and if \vec{M} is piecewise constant the magnetization term will disappear in equation analysis. It is necessary to reformulate the magnetic term so that it can be incorporated into the divergence. We have from (2.5)

$$(2.14) \quad \iiint_V \nabla \times \vec{M} dV = \oint_S \vec{n} \times \vec{M} dS.$$

Magnetic terms that will obey

$$(2.15) \quad \vec{n} \times \vec{M} = \vec{n} \cdot \vec{N}$$

can be formed by defining \vec{N} as the antisymmetric dyadic

$$\vec{N} = \begin{pmatrix} 0 & M_z & -M_y \\ -M_z & 0 & M_x \\ M_y & -M_x & 0 \end{pmatrix}$$

Using this relation, we can write eq. (2.13) as

$$(2.16) \quad \nabla \cdot \left(\frac{\nabla \vec{A}}{\mu} + \vec{N} \right) + \vec{J} = 0$$

This follows because integration by parts will produce surface terms $\vec{n} \cdot \vec{N}$, which are equivalent to the required surface terms $\nabla \times \vec{M}$.

Expanded in Cartesian coordinates, this results in the three equations

$$(2.17) \quad \begin{aligned} \nabla \cdot \left(\frac{\nabla A_x}{\mu} + \vec{N}_x \right) + J_x &= 0 \\ \nabla \cdot \left(\frac{\nabla A_y}{\mu} + \vec{N}_y \right) + J_y &= 0 \\ \nabla \cdot \left(\frac{\nabla A_z}{\mu} + \vec{N}_z \right) + J_z &= 0 \end{aligned}$$

where the \vec{N}_i are the rows of \vec{N}

In this formulation, the Natural boundary condition will be defined as the value of the normal component of the argument of the divergence, eg.

$$(2.18) \quad \text{natural}(A_x) = \vec{n} \cdot \left(\frac{\nabla A_x}{\mu} + \vec{N}_x \right).$$

As an example, we will compute the magnetic field in a generic magnetron. In this case, only M_z is applied by the magnets, and as a result A_z will be zero. We will therefore delete A_z from the analysis. The outer and inner magnets are in reversed orientation, so the applied M_z is reversed in sign.

Descriptor 2.3: 3D Magnetron.pde

```

TITLE 'Oval Magnet '

COORDINATES
  CARTESIAN3

SELECT
  alias(x) = "X(cm)"
  alias(y) = "Y(cm)"
  alias(z) = "Z(cm)"
  nodelimit = 40000
  errlim=1e-4

VARIABLES
  Ax,Ay          { assume Az is zero! }

DEFINITIONS
  MuMag=1.0              { Permeabilities: }
  MuAir=1.0
  MuSST=1000
  MuTarget=1.0
  Mu=MuAir              { default to Air }

  MzMag = 10000          { permanent magnet strength }
  Mz = 0
  Nx = vector(0,Mz,0)
  Ny = vector(-Mz,0,0)

  B = curl(Ax,Ay,0)      { magnetic flux density }
  Bxx= -dz(Ay)
  Byy= dz(Ax)            { "By" is a reserved word. }
  Bzz= dx(Ay)-dy(Ax)

EQUATIONS
  div(grad(Ax)/mu + Nx) = 0
  div(grad(Ay)/mu + Ny) = 0

EXTRUSION
  SURFACE "Boundary Bottom"          Z=-5
  SURFACE "Magnet Plate Bottom"      Z=0
  LAYER "Magnet Plate"
  SURFACE "Magnet Plate Top"         Z=1
  LAYER "Magnet"
  SURFACE "Magnet Top"               Z=2
  SURFACE "Boundary Top"             Z=8

BOUNDARIES
  Surface "boundary bottom"
    value (Ax)=0 value (Ay)=0
  Surface "boundary top"
    value (Ax)=0 value (Ay)=0

REGION 1 { Air bounded by conductive box }
  START (20,-10)
    value (Ax)=0 value (Ay)=0
    arc(center=20,0) angle=180

```

```

Line TO (-20,10)
arc(center=-20,0) angle=180
LINE TO FINISH

REGION 2 { Magnet Plate Perimeter and outer magnet }
LAYER "Magnet Plate"
  Mu=MuSST
LAYER "Magnet"
  Mu=MuMag
  Mz=MzMag
START (20,-8)
  arc(center=20,0) angle=180
  Line TO (-20,8)
  arc(center=-20,0) angle=180
  LINE TO FINISH

REGION 3 {Air }
LAYER "Magnet Plate"
  Mu=MuSST
START (20,-6)
  arc(center=20,0) angle=180
  Line TO (-20,6)
  arc(center=-20,0) angle=180
  LINE TO FINISH



REGION 4 {Inner Magnet }
LAYER "Magnet Plate"
  Mu=MuSST
LAYER "Magnet"
  Mu=MuMag
  Mz=-MzMag
START (20,-2)
  arc(center=20,0) angle=180
  Line TO (-20,2)
  arc(center=-20,0) angle=180
  LINE TO FINISH


MONITORS
grid(x,z) on y=0
grid(x,y) on z=1.01
grid(x,z) on y=1

PLOTS
grid(x,y) on z=1.01
grid(y,z) on x=0
grid(x,z) on y=0
contour(Ax) on x=0
contour(Ay) on y=0
vector(Bxx,Byy) on z=2.01 norm
vector(Byy,Bzz) on x=0 norm
vector(Bxx,Bzz) on y=4 norm
contour(magnitude(Bxx,Byy,Bzz)) on z=2.01 LOG

```

END





Chapter 3

Waveguides

A waveguide is any of several kinds of structure intended to direct the propagation of high-frequency electromagnetic energy along specific paths. While the analysis of bends and terminations in such a system is an essentially three-dimensional problem, the propagation in long straight segments of the guide can be reduced to a two-dimensional analysis. In this case, we assume that the guide is of uniform cross-section in the (X,Y) plane, unvarying in the Z-dimension of the propagation direction. In this configuration, we can make the assumption that the fields inside the guide may be represented as a sinusoidal oscillation in time and space, and write

$$(3.1) \quad \begin{aligned} \vec{E}(x, y, z, t) &= \vec{\mathcal{E}}(x, y) \exp(i\omega t - i\gamma z) \\ \vec{H}(x, y, z, t) &= \vec{\mathcal{H}}(x, y) \exp(i\omega t - i\gamma z) \end{aligned}$$

It is easy to see that these expressions describe a traveling wave, since the imaginary exponential generates sines and cosines, and the value of the exponential will be the same wherever $\gamma z = \omega t$. A purely real γ implies an unattenuated propagating mode with wavelength $\lambda = 2\pi / \gamma$ along the z direction.

We start from the time-dependent form of Maxwell's equations

$$(3.2) \quad \begin{aligned} \nabla \times \vec{H} &= \vec{J} + \frac{\partial \vec{D}}{\partial t} = \vec{J} + \epsilon \frac{\partial \vec{E}}{\partial t} \\ \nabla \cdot \vec{B} &= \nabla \cdot (\mu \vec{H}) = 0 \\ \nabla \times \vec{E} &= -\frac{\partial \vec{B}}{\partial t} = -\mu \frac{\partial \vec{H}}{\partial t} \\ \nabla \cdot \vec{D} &= \nabla \cdot (\epsilon \vec{E}) = \rho \end{aligned}$$

Assume then that $\vec{J} = 0$ and $\rho = 0$, and apply (3.1):

$$(3.3) \quad \begin{aligned} \nabla \times \vec{\mathcal{H}} &= i\omega\epsilon\vec{\mathcal{E}} & \nabla \cdot (\mu\vec{\mathcal{H}}) &= 0 \\ \nabla \times \vec{\mathcal{E}} &= -i\omega\mu\vec{\mathcal{H}} & \nabla \cdot (\epsilon\vec{\mathcal{E}}) &= 0 \end{aligned}$$

Taking the curl of each curl equation in (3.3) and substituting gives

$$(3.4) \quad \begin{aligned} \nabla \times \left(\frac{\nabla \times \vec{\mathcal{H}}}{\epsilon} \right) &= \omega^2 \mu \vec{\mathcal{H}} \\ \nabla \times \left(\frac{\nabla \times \vec{\mathcal{E}}}{\mu} \right) &= \omega^2 \epsilon \vec{\mathcal{E}} \end{aligned}$$

In view of (3.1), we can write

$$(3.5) \quad \begin{aligned} \nabla &= \vec{1}_x \frac{\partial}{\partial x} + \vec{1}_y \frac{\partial}{\partial y} - \vec{1}_z i\gamma \\ &= \nabla_T - \vec{1}_z i\gamma \end{aligned}$$

with ∇_T denoting the operator in the transverse plane.

Homogeneous Waveguides

In many cases, the waveguide under analysis consists of a metal casing, either empty or filled homogeneously with an isotropic dielectric. In these cases, the analysis can be simplified.

Eq. (3.3) can be expanded using (3.5) and rearranged to express the transverse x and y components in terms of the axial z components \mathcal{H}_z and \mathcal{E}_z .

$$(3.6) \quad \begin{aligned} (\omega^2 \mu \epsilon - \gamma^2) \mathcal{E}_x &= -i \left(\omega \mu \frac{\partial \mathcal{H}_z}{\partial y} + \gamma \frac{\partial \mathcal{E}_z}{\partial x} \right) \\ (\omega^2 \mu \epsilon - \gamma^2) \mathcal{E}_y &= i \left(\omega \mu \frac{\partial \mathcal{H}_z}{\partial x} - \gamma \frac{\partial \mathcal{E}_z}{\partial y} \right) \\ (\omega^2 \mu \epsilon - \gamma^2) \mathcal{H}_x &= i \left(\omega \epsilon \frac{\partial \mathcal{E}_z}{\partial y} - \gamma \frac{\partial \mathcal{H}_z}{\partial x} \right) \\ (\omega^2 \mu \epsilon - \gamma^2) \mathcal{H}_y &= -i \left(\omega \epsilon \frac{\partial \mathcal{E}_z}{\partial x} + \gamma \frac{\partial \mathcal{H}_z}{\partial y} \right) \end{aligned}$$

The i in the right hand side corresponds to a phase shift of $\pi/2$ in the expansion (3.1)

Applying (3.5), the divergence equations of (3.3) become

$$(3.7) \quad \begin{aligned} i\gamma \mathcal{H}_z &= \frac{\partial \mathcal{H}_x}{\partial x} + \frac{\partial \mathcal{H}_y}{\partial y} \\ i\gamma \mathcal{E}_z &= \frac{\partial \mathcal{E}_x}{\partial x} + \frac{\partial \mathcal{E}_y}{\partial y} \end{aligned}$$

so the z component equations of (3.4) are

$$(3.8) \quad \begin{aligned} \nabla_T \cdot (\nabla_T \mathcal{H}_z) + (\omega^2 \mu \epsilon - \gamma^2) \mathcal{H}_z &= 0 \\ \nabla_T \cdot (\nabla_T \mathcal{E}_z) + (\omega^2 \mu \epsilon - \gamma^2) \mathcal{E}_z &= 0 \end{aligned}$$

These are eigenvalue equations in \mathcal{E}_z and \mathcal{H}_z , and the values of

$(\omega^2 \mu \epsilon - \gamma^2)$ for which solutions exist constitute the propagation

constants of the unattenuated propagation modes that can be supported in the guide under analysis. For any eigenvalue, there are an infinite number of combinations of $\omega, \epsilon, \mu, \gamma$ which can excite this mode, and the exact determination will depend on the materials and the driving frequency.

TE and TM modes

In a homogeneously filled waveguide, there exist two sets of distinct modes. One set of modes has no magnetic field component in the propagation direction, and are referred to as Transverse Magnetic, or TM, modes. The other set has no electric field component in the propagation direction, and are referred to as Transverse Electric, or TE, modes. In either case, one member of (3.8) vanishes, leaving only a single variable and a single equation. Correspondingly, equations (3.6) are simplified by the absence of one or the other field component.

In the TM case, we have $\mathcal{E}_z = 0$, and the first of (3.8)

$$(3.9) \quad \nabla_T \cdot (\nabla_T \mathcal{H}_z) + (\omega^2 \mu \epsilon - \gamma^2) \mathcal{H}_z = 0$$

The boundary condition at an electrically conducting wall is $\hat{n} \cdot \vec{H} = 0$, Through (3.6), this implies $\hat{n} \cdot \nabla_T \mathcal{H}_z = 0$, which is the Natural boundary condition of (3.9).

In the TE case, we have $\mathcal{H}_z = 0$, and the second of (3.8)

$$(3.10) \quad \nabla_T \cdot (\nabla_T \mathcal{E}_z) + (\omega^2 \mu \epsilon - \gamma^2) \mathcal{E}_z = 0.$$

The boundary condition at a metallic wall is $\hat{n} \times \vec{E} = 0$, which requires that tangential components of \vec{E} be zero in the wall. Since \mathcal{E}_z is always tangential to the wall, the boundary condition is the Dirichlet condition $\mathcal{E}_z = 0$.

In the following example, we compute the first few TE modes of a waveguide of complex cross-section. The natural boundary condition allows an infinite number of solutions, differing only by a constant offset in the eigenfunction, so we add an integral constraint to center the eigenfunctions around zero. Since all the material parameters are contained in the eigenvalue, it is unnecessary to concern ourselves with their values. Likewise, the computation of the transverse field components are scaled by constants, but the shapes are unaffected.

Descriptor 3.1 Waveguide.pde

```

title "TE Waveguide"

select
  modes = 4    { This is the number of Eigenvalues desired. }

variables
  Hz

definitions
  L = 2
  h = 0.5      ! half box height
  g = 0.01     ! half-guage of wall
  s = 0.3*L    ! septum depth
  tang = 0.1   ! half-width of tang
  Hx = -dx(Hz)
  Hy = -dy(Hz)
  Ex = Hy
  Ey = -Hx

equations
  div(grad(Hz)) + lambda*Hz = 0

```

```

constraints      { since Hz has only natural boundary
                  conditions, we need to constrain the answer }
  integral(Hz) = 0


boundaries
  region 1
    start(0,0)
    natural(Hz) = 0
    line to (L,0) to (L,1) to (0,1) to (0,h+g)
    natural(Hz) = 0
    line to (s-g,h+g) to (s-g,h+g+tang) to (s+g,h+g+tang)
      to (s+g,h-g-tang) to (s-g,h-g-tang)
      to (s-g,h-g) to (0,h-g)
    to finish



monitors
  contour(Hz)

plots
  contour(Hz) painted
  vector(Hx,Hy) as "Transverse H" norm
  vector(Ex,Ey) as "Transverse E" norm

end 13474

```





Non-Homogeneous Waveguides²

In many applications, a waveguide is partially or inhomogeneously filled with dielectric material. In this case, pure TE and TM modes do not exist. Both \mathcal{E}_z and \mathcal{H}_z exist simultaneously, and the propagation modes are hybrid in nature.

It is possible to address a simultaneous solution of equations (3.4) in a manner similar to (3.8). However, care must be taken to keep the ε parameter inside of some of the derivatives, and problems arise with the simplifications implicit in (3.7). This approach also has been plagued with spurious solution modes. It is claimed that these spurious modes arise because the axial field model does not explicitly impose $\nabla \cdot \vec{B} = 0$, and that the spurious modes are those for which this condition is violated.

An alternative approach seeks to reduce the equations (3.4) to a pair of equations in the transverse components of the magnetic field, $\mathcal{H}_T = \mathcal{H}_x \hat{1}_x + \mathcal{H}_y \hat{1}_y$. In the process, the condition $\nabla \cdot \vec{B} = 0$ is explicitly imposed, and it is claimed that no spurious modes arise.

In the development that follows, we continue to treat μ as a constant (invalidating use where magnetic materials are present), but we exercise more care in the treatment of ε .

For notational convenience, we will denote the components of $\vec{\mathcal{H}}$ as $\vec{\mathcal{H}} = a \hat{1}_x + b \hat{1}_y + c \hat{1}_z$ and use subscripts to denote differentiation. The first equation of (3.4) can then be expanded with (3.5) to give

$$\begin{aligned}
 (b_x / \varepsilon)_y - (a_y / \varepsilon)_x - i\gamma c_x / \varepsilon + \gamma^2 a / \varepsilon &= \omega^2 \mu a \\
 (a_y / \varepsilon)_x - (b_x / \varepsilon)_y - i\gamma c_y / \varepsilon + \gamma^2 b / \varepsilon &= \omega^2 \mu b \\
 -(c_x / \varepsilon)_x - (c_y / \varepsilon)_y - i\gamma (a / \varepsilon)_x - i\gamma (b / \varepsilon)_y &= \omega^2 \mu c
 \end{aligned}
 \tag{3.11}$$

The condition $\nabla \cdot \vec{B} = 0$ allows us to replace

² The development given here follows that of Fernandez and Lu, "Microwave and Optical Waveguide Analysis", and of Silvester and Ferrari, "Finite Elements for Electrical Engineers".

$$(3.12) \quad i\gamma c = a_x + b_y$$

and to eliminate the third equation. We can also define $\varepsilon = \varepsilon_r \varepsilon_0$ and $\mu = \mu_0$ and multiply through by ε_0 leaving

$$(3.13) \quad \begin{aligned} (b_x / \varepsilon_r)_y - (a_y / \varepsilon_r)_y - (a_x + b_y)_x / \varepsilon_r + \gamma^2 a / \varepsilon_r &= \omega^2 \mu_0 \varepsilon_0 a \\ (a_y / \varepsilon_r)_x - (b_x / \varepsilon_r)_x - (a_x + b_y)_y / \varepsilon_r + \gamma^2 b / \varepsilon_r &= \omega^2 \mu_0 \varepsilon_0 b \end{aligned}$$

In vector form we can write this as

$$(3.14) \quad \nabla_T \times \left(\frac{\nabla_T \times \vec{\mathcal{H}}_T}{\varepsilon_r} \right) - \frac{\nabla_T (\nabla_T \cdot \vec{\mathcal{H}}_T)}{\varepsilon_r} + \frac{\gamma^2 \vec{\mathcal{H}}_T}{\varepsilon_r} = \omega^2 \mu_0 \varepsilon_0 \vec{\mathcal{H}}_T$$

The equation pair (3.13) is an eigenvalue problem in γ^2 . We can no longer bundle the ω^2 and γ^2 terms inside the eigenvalue, because the ε_r dividing γ^2 is now variable across the domain. Given a driving frequency ω , we can compute the axial wave numbers γ for which propagating modes exist.

Boundary Conditions

To see what the natural boundary conditions imply, integrate the second order terms of (3.13) by parts:

$$(3.15) \quad \begin{aligned} &\iint_T \left[(b_x / \varepsilon_r)_y - (a_y / \varepsilon_r)_y - (a_x + b_y)_x / \varepsilon_r \right] dx dy \\ &\longrightarrow \oint \left[n_y (b_x - a_y) / \varepsilon_r - n_x (a_x + b_y) / \varepsilon_r \right] dl \\ &\iint_T \left[(a_y / \varepsilon_r)_x - (b_x / \varepsilon_r)_x - (a_x + b_y)_y / \varepsilon_r \right] dx dy \\ &\longrightarrow \oint \left[n_x (a_y - b_x) / \varepsilon_r - n_y (a_x + b_y) / \varepsilon_r \right] dl \end{aligned}$$

We have shown only the contour integrals arising from the integration, and suppressed the area integral correcting for varying ε . This term will be correctly added by FlexPDE, and does not contribute to the boundary condition.

The integrand of the contour integrals is the value represented by the natural boundary condition statement in FlexPDE.

The boundary conditions which must be satisfied at an electrically conducting wall are

$$(3.16) \quad \begin{aligned} \hat{n} \cdot \vec{\mathcal{H}} &= 0 \\ \hat{n} \times \vec{\mathcal{E}} &= 0 \end{aligned}$$

The first condition requires that $n_x \mathcal{H}_x + n_y \mathcal{H}_y + n_z \mathcal{H}_z = 0$. At a vertical wall, $n_y = n_z = 0$, and the condition becomes simply $\mathcal{H}_x = 0$.

Similarly, at a horizontal wall, it is $\mathcal{H}_y = 0$. Both are easily expressed as Value boundary conditions. At an oblique wall, the condition can be expressed as an implicit value boundary condition for one of the components.

The second condition requires that the tangential components of $\vec{\mathcal{E}}$ must vanish in the wall. In particular, \mathcal{E}_z is always tangential and must therefore be zero. From (3.3) we can derive $i\omega \varepsilon \mathcal{E}_z = (b_x - a_y)$. But this is just the first term of the integrands in (3.15), so at a vertical wall we can set $\text{Natural}(\mathcal{H}_y) = 0$, and at a horizontal wall we can use

$\text{Natural}(\mathcal{H}_x) = 0$. These are the reverse assignments from the value conditions above, so the two form a complementary set and completely specify the boundary conditions for (3.13). Similar arguments can be used at a magnetic wall, resulting in a reversed assignment of value and natural boundary conditions.

Material Interfaces

At a material interface, Maxwell's equations require that the tangential components of $\vec{\mathcal{E}}$ and $\vec{\mathcal{H}}$ and the normal components of $\varepsilon \vec{\mathcal{E}}$ and $\mu \vec{\mathcal{H}}$ must be continuous.

The tangential continuity of components $\mathcal{H}_x = a$ and $\mathcal{H}_y = b$ is automatically satisfied, because FlexPDE stores only a single value of variables at the interface. Continuity of $\mathcal{H}_z = c$, which is always

tangential, requires, using (3.12), $(a_x + b_y)_1 = (a_x + b_y)_2$. Continuity of \mathcal{E}_z requires $(b_x - a_y)_1 = (b_x - a_y)_2$.

Now consider the integrals (3.15) to be taken over each material independently. Each specifies in a general sense the “flux” of some quantity outward from the region. The sum of the two integrands, taking into account the reversed sign of the outward normal, specifies the conservation of the “flux”. In the usual case, the sum is zero, representing “flux” conservation. In our case, we must specify a jump in the flux in order to satisfy the requirements of Maxwell’s equations.

For the \mathcal{H}_x component equation we have, using the outward normals from region 1,

$$\text{integrand}_1 + \text{integrand}_2 = n_y \left[\left(\frac{b_x - a_y}{\epsilon_r} \right)_1 - \left(\frac{b_x - a_y}{\epsilon_r} \right)_2 \right] - n_x \left[\left(\frac{a_x + b_y}{\epsilon_r} \right)_1 - \left(\frac{a_x + b_y}{\epsilon_r} \right)_2 \right]$$

But the continuity requirements above dictate that the numerators be continuous, so the internal natural boundary condition becomes

$$\text{integrand}_1 + \text{integrand}_2 = \left[n_y (b_x - a_y) - n_x (a_x + b_y) \right] \left[\left(\frac{1}{\epsilon_r} \right)_1 - \left(\frac{1}{\epsilon_r} \right)_2 \right]$$

By a similar argument, the internal natural boundary condition for the \mathcal{H}_y component equation is

$$\text{integrand}_1 + \text{integrand}_2 = \left[n_x (a_x - b_y) - n_y (a_x + b_y) \right] \left[\left(\frac{1}{\epsilon_r} \right)_1 - \left(\frac{1}{\epsilon_r} \right)_2 \right]$$

Clearly, at an internal interface where ϵ_r is continuous, the internal natural boundary condition reduces to zero, which is the default condition.

In the example which follows, we consider a simple 2x1 metal box with dielectric material in the left half. Note that FlexPDE will compute the eigenvalues with lowest magnitude, regardless of sign, while negative

eigenvalues correspond to modes with propagation constants below cutoff, and are therefore not physically realizable.

Descriptor 3.2 Filledguide.pde

```

title "Filled Waveguide"

select
  modes = 8      { This is the number of Eigenvalues desired. }

variables
  hx,hy

definitions
  cm = 0.01      ! conversion from cm to meters
  b = 1*cm       ! box height
  L = 2*b        ! box width
  epsr
  epsr1=1        epsr2=1.5
  ejump = 1/epsr2-1/epsr1 ! the boundary jump parameter
  eps0 = 8.85e-12
  mu0 = 4e-7*pi
  c = 1/sqrt(mu0*eps0) ! light speed
  k0b = 4
  k0 = k0b/b
  k02 = k0^2      ! k0^2=omega^2*mu0*eps0

  curlh = dx(Hy)-dy(Hx) ! terms used in equations and BC's
  divh = dx(Hx)+dy(Hy)

equations
  dx(divh)/epsr - dy(curlh/epsr) + k02*Hx - lambda*Hx/epsr = 0
  dx(curlh/epsr) + dy(divh)/epsr + k02*Hy - lambda*Hy/epsr = 0

boundaries
  region 1 epsr=epsr1
    start(0,0)
    natural(Hx) = 0 value(Hy)=0
    line to (L,0)
    value(Hx) = 0 value(Hy)=0 natural(Hy)=0
    line to (L,b)
    natural(Hx) = 0 value(Hy)=0
    line to (0,b)
    value(Hx) = 0 natural(Hy)=0
    line to finish

  region 2 epsr=epsr2
    start(b,b)
    line to (0,b) to (0,0) to (b,0)
    natural(Hx) = normal(-ejump*divh,ejump*curlh)
    natural(Hy) = normal(-ejump*curlh,-ejump*divh)
    line to finish

monitors
  contour(Hx) range=(-3,3)
  contour(Hy) range=(-3,3)

```



```

plots
  contour(Hx) range=(-3,3) report(k0b)
  report(sqrt(abs(lambda))/k0)
  surface(Hx) range=(-3,3) report(k0b)
  report(sqrt(abs(lambda))/k0)
  contour(Hy) range=(-3,3) report(k0b)
  report(sqrt(abs(lambda))/k0)
  surface(Hy) range=(-3,3) report(k0b)
  report(sqrt(abs(lambda))/k0)




```

```

summary export
  report(k0b)
  report lambda
  report(sqrt(abs(lambda))/k0)

```

```
end
```



Revision Log

Rev 1, 11/24/02

Corrected errors in listing of Descriptor 2.3, "3D_Magnetron.pde".

References

N. J. Cronin, "Microwave and Optical Waveguides", London, Institute of Physics Publishing, 1995.

F. Anibal Fernandez and Yilong Lu, "Microwave and Optical Waveguide Analysis", Somerset, UK, Research Studies Press, Ltd. 1996.

S. R. H. Hoole, "Computer-Aided Analysis and Design of Electromagnetic Devices", New York, Elsevier, 1989.

Nathan Ida and Joao P.A. Bastos "Electromagnetics and Calculation of Fields", New York, Springer-Verlag, 1992.

J. D. Jackson, "Classical Electrodynamics", Second Edition, New York, John Wiley & Sons, 1975.

Jianming Jin, "The Finite Element Method in Electromagnetics", New York, John Wiley & Sons, Inc, 1993.

Peter P. Silvester and Ronald L. Ferrari, "Finite Elements for Electrical Engineers", Third Edition, Cambridge University Press, 1996.

C. T. Tai, "Generalized Vector and Dyadic Analysis", New York, IEEE Press, 1992.